Translation and Dilation (Scaling)

Defining and plotting a function

In[1]:=
```
f[t_] := Piecewise[{{ Sin[t^2], 0 ≤ t < Sqrt[3 * Pi]}}]
```

In[2]:=
```
Plot[f[t], {t, -2, 5}, Filling -> Axis,
 PlotStyle -> Black, PlotLabels → Placed[Automatic, Below]]
```

Out[2]=



Translation

In[3]:=
```
T[f_, x_, b_] := f[x - b]
```

In[4]:=
```
T[f, x, 1]
```

Out[4]=

$$\begin{cases} \sin\left[(-1+x)^2\right] & 0 \le -1+x < \sqrt{3\,\pi} \\ 0 & \text{True} \end{cases}$$

In[5]:=
```
Plot[T[f, x, 1], {x, -2, 5}, Filling -> Axis,
 PlotStyle -> Red, PlotLabels → Placed[Automatic, Below]]
```

Out[5]=



In[6]:=
```
Plot[T[f, x, -1], {x, -2, 5}, Filling -> Axis,
 PlotStyle -> Blue, PlotLabels → Placed[Automatic, Below]]
```

Out[6]=



In[7]:=
```
Plot[{f[x], T[f, x, 2], T[f, x, -2]}, {x, -2, 6},
 Filling -> Axis, PlotStyle -> {Black, Red, Blue},
 PlotRange -> All, PlotLabels → Placed[Automatic, Below]
]
```

Out[7]=



Dilation (Scaling)

In[8]:=
```
S[f_, t_, a_] := Sqrt[a] * f[a * t]
```

In[9]:=
```
Plot[S[f, t, 1/2], {t, -1, 7}, Filling -> Axis,
 PlotStyle -> Red, PlotLabels → Placed[Automatic, Below]]
```

Out[9]=



In[10]:=
```
Plot[S[f, t, 2], {t, -1, 7}, Filling -> Axis,
 PlotStyle -> Blue, PlotLabels → Placed[Automatic, Below]]
```

Out[10]=



In[11]:=
```
Plot[{f[t], S[f, t, 1/2], S[f, t, 2]}, {t, -1, 7},
 Filling -> Axis, PlotStyle -> {Black, Red, Blue},
 PlotRange -> All, PlotLabels → Placed[Automatic, Below]
]
```

Out[11]=



Translation and scaling do not commute!

In[12]:=
```
s[t_] = S[f, t, 2]
```

Out[12]= $\sqrt{2}\left(\begin{cases} \text{Sin}\left[4\,t^2\right] & 0 \le 2\,t < \sqrt{3\,\pi} \\ 0 & \text{True} \end{cases}\right)$

In[13]:=
```
u[t_] = T[f, t, -1]
```

Out[13]= $\begin{cases} \text{Sin}\left[(1+t)^2\right] & 0 \le 1+t < \sqrt{3\,\pi} \\ 0 & \text{True} \end{cases}$

first scaling (by 2), then translation (by -1)

In[14]:=
```
TS = T[s, t, -1]
```

Out[14]= $\sqrt{2}\left(\begin{cases} \text{Sin}\left[4\,(1+t)^2\right] & 0 \le 2\,(1+t) < \sqrt{3\,\pi} \\ 0 & \text{True} \end{cases}\right)$

first translation (by -1), then first scaling (by 2)

In[15]:=
```
ST = S[u, t, 2]
```

Out[15]= $\sqrt{2}\left(\begin{cases} \text{Sin}\left[(1+2\,t)^2\right] & 0 \le 1+2\,t < \sqrt{3\,\pi} \\ 0 & \text{True} \end{cases}\right)$

In[16]:=
```
Plot[{TS, ST}, {t, -1, 2}, Filling -> Axis, PlotStyle -> {Green, Blue},
 PlotRange -> All, PlotLabels -> Placed[Automatic, Below]]
```

Out[16]=



Integrals

In[17]:=
```
NIntegrate[{f[t], T[f, t, 1], T[f, t, -1]}, {t, -2, 6}]
```

Out[17]=
```
{0.788259, 0.788259, 0.788259}
```

In[18]:=
```
NIntegrate[{f[t]^2, T[f, t, 1/2]^2, T[f, t, -1]^2}, {t, -2, 6}]
```

Out[18]=
```
{1.3145, 1.3145, 1.3145}
```

In[19]:=
```
NIntegrate[{f[t], S[f, t, 1/2], S[f, t, 2]}, {t, 0, 10}]
```

Out[19]=
```
{0.788259, 1.11477, 0.557383}
```

In[20]:=
```
NIntegrate[{f[t]^2, S[f, t, 1/2]^2, S[f, t, 2]^2}, {t, 0, 10}]
```

Out[20]=
```
{1.3145, 1.3145, 1.3145}
```